Systems of Linear Equations

Graphical Representation of Linear Equations

```
eqns = {x1 + 2 x2 == 5,
        3 x1 + 4 x2 == 6};

soln = Table[Solve[eqns[[k]], x2],
   {k, 1, 2}] // Flatten
```

$\left\{x2 \to \frac{5 - x1}{2},\ x2 \to -\frac{3}{4}\ (-2 + x1)\right\}$

```
lines = Table[x2 /. soln[[k]],
  {k, 1, 2}]
```

$\left\{\frac{5 - x1}{2},\ -\frac{3}{4}\ (-2 + x1)\right\}$

```
Plot[Evaluate[lines], {x1, -8, 5},
  AxesLabel → {x1, x2},
  PlotStyle → {Orange, ForestGreen}];
```



The Number of Solutions of a System of Linear Equations

The graphical representation of linear equations makes it clear that there are precisely three possibilities for the solution sets of such systems:
(1) There is no solution
(2) There is precisely one solution
(1) There is an infinite number of solutions

```
(* No solution -- parallel lines *)

eqns = {3 x1 - 2 x2 == 12,
        3 x1 - 2 x2 == 6};

soln = Table[Solve[eqns[[k]], x2], {k, 1, 2}] // Flatten
lines = Table[x2 /. soln[[k]], {k, 1, 2}];

Plot[Evaluate[lines], {x1, -2, 8},
  PlotLabel -> "No solution",
  AxesLabel → {x1, x2},
  PlotStyle → {Orange, ForestGreen}];
```

$\left\{x2 \to \frac{3}{2}\,(-4 + x1),\ x2 \to \frac{3}{2}\,(-2 + x1)\right\}$



```
(* One solution -- intersecting lines *)

eqns = {3 x1 - 2 x2 == 12,
        2 x1 + x2 == 6};

soln = Table[Solve[eqns[[k]], x2], {k, 1, 2}] // Flatten
lines = Table[x2 /. soln[[k]], {k, 1, 2}];

Plot[Evaluate[lines], {x1, -2, 8},
  PlotLabel -> "One solution",
  AxesLabel → {x1, x2},
  PlotStyle → {Orange, ForestGreen}];
```

$\left\{x2 \to \frac{3}{2}\,(-4 + x1),\ x2 \to -2\,(-3 + x1)\right\}$



```
(* Infinite number of solutions -- coincident lines *)

eqns = {3 x1 - 2 x2 == 12,
        6 x1 - 4 x2 == 24};

soln = Table[Solve[eqns[[k]], x2], {k, 1, 2}] // Flatten
lines = Table[x2 /. soln[[k]], {k, 1, 2}];

Plot[Evaluate[lines], {x1, -2, 8},
  PlotLabel -> "Infinite number of solutions",
  AxesLabel → {x1, x2},
  PlotStyle → {Orange, ForestGreen}];
```

$\left\{x2 \to \frac{3}{2}\,(-4 + x1),\ x2 \to \frac{3}{2}\,(-4 + x1)\right\}$



Solving Systems of Linear Equations

Using Mathematica to solve linear equations

```
eqns = {x1 + 3 x2 == 4,
        2 x1 - x2 == 5};

Solve[eqns, {x1, x2}] // Flatten
```

$\left\{x1 \to \frac{19}{7},\ x2 \to \frac{3}{7}\right\}$

■ Lay Example 1.1.1, p.5

```
eqns = {x1 - 2 x2 + x3 == 0,
            2 x2 - 8 x3 == 8,
      -4 x1 + 5 x2 + 9 x3 == -9};

Solve[eqns, {x1, x2, x3}] // Flatten

{x1 → 29, x2 → 16, x3 → 3}
```

Linear Equations to Matrices

```
<< LinearAlgebra`MatrixManipulation`
```

```
Clear[x, y, x1, x2, x3];

eqns = {a11 x + a12 y == c1,
        a21 x + a22 y == c2};

LinearEquationsToMatrices[eqns, {x, y}];
Map[MatrixForm, %]
```

$$\left\{\begin{pmatrix} a11 & a12 \\ a21 & a22 \end{pmatrix}, \begin{pmatrix} c1 \\ c2 \end{pmatrix}\right\}$$

```
eqns = {x1 + 2 x2 == 5,
       3 x1 + 4 x2 == 6};

LinearEquationsToMatrices[eqns, {x1, x2}];
Map[MatrixForm, %]
```

$$\left\{\begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}, \begin{pmatrix} 5 \\ 6 \end{pmatrix}\right\}$$

```
eqns = {x1 - 2 x2 + x3 == 0,
             2 x2 - 8 x3 == 8,
      -4 x1 + 5 x2 + 9 x3 == -9};

LinearEquationsToMatrices[eqns, {x1, x2, x3}];
Map[MatrixForm, %]
```

$$\left\{\begin{pmatrix} 1 & -2 & 1 \\ 0 & 2 & -8 \\ -4 & 5 & 9 \end{pmatrix}, \begin{pmatrix} 0 \\ 8 \\ -9 \end{pmatrix}\right\}$$

Matrices to Linear Equations

```
matricesToLinearEquations[{m_, b_}, vars_] :=
 If[
  Dimensions[m][[2]] == Length[vars] && Dimensions[m][[1]] == Length[b],
  Thread[m.vars == b],
  "Error: matricesToLinearEquations: Sizes don't match."]
```

```
m = (1  2
     3  4);
b = {5, 6};
vars = {x, y};

matricesToLinearEquations[{m, b}, vars]
```

{x + 2 y == 5, 3 x + 4 y == 6}

```
eqns = {x1 - 2 x2 + x3 == 0,
             2 x2 - 8 x3 == 8,
      -4 x1 + 5 x2 + 9 x3 == -9};

vars = {x1, x2, x3};

matricesToLinearEquations[
 LinearEquationsToMatrices[eqns, vars],
 vars]
```

{x1 - 2 x2 + x3 == 0, 2 x2 - 8 x3 == 8, -4 x1 + 5 x2 + 9 x3 == -9}

Augmented Matrices

Make an augmented matrix.

```
<< LinearAlgebra`MatrixManipulation`
```

```
makeAug::usage = "makeAug[{mat,vec}] takes a pair consisting of a
    matrix and vector, and returns the associated augmented matrix.";

makeAug[{mat_, vec_}] :=
 If[Length[mat] == Length[vec],
  AppendRows[mat, Map[List, vec]],
  "Error: makeAug: Sizes don't match."]
```

```
eqns = {x1 - 2 x2 + x3 == 0,
          2 x2 - 8 x3 == 8,
   -4 x1 + 5 x2 + 9 x3 == -9};

mv = LinearEquationsToMatrices[eqns, {x1, x2, x3}];

aug = makeAug[mv];
% // MatrixForm
```

$$\begin{pmatrix} 1 & -2 & 1 & 0 \\ 0 & 2 & -8 & 8 \\ -4 & 5 & 9 & -9 \end{pmatrix}$$

Decompose an augmented matrix

```
decomposeAug::usage = "decomposeAug[aug] decomposes an augmented matrix
    into a pair containing its component matrix and vector.";

decomposeAug[aug_] :=
 {Table[Drop[aug[[k]], -1], {k, Length[aug]}],
  Table[aug[[k, -1]], {k, Length[aug]}]}
```

```
a = ( 1  -2   1   0
      0   2  -8   8
     -4   5   9  -9 );

decomposeAug[a];
Map[MatrixForm, %]
```

$$\left\{\begin{pmatrix} 1 & -2 & 1 \\ 0 & 2 & -8 \\ -4 & 5 & 9 \end{pmatrix}, \begin{pmatrix} 0 \\ 8 \\ -9 \end{pmatrix}\right\}$$

Using RowReduce to Solve Linear Equations

■ Lay Example 1.1.1, p.5

```
<< LinearAlgebra`MatrixManipulation`
```

```
rr = RowReduce[aug];
% // MatrixForm

soln = TakeColumns[rr, -1] // Flatten

ToRules[{x1, x2, x3} == soln]
```

$$\begin{pmatrix} 1 & 0 & 0 & 29 \\ 0 & 1 & 0 & 16 \\ 0 & 0 & 1 & 3 \end{pmatrix}$$

```
{29, 16, 3}
```

```
{x1 → 29, x2 → 16, x3 → 3}
```

Detecting an Inconsistent System

■ Lay Example 1.1.1, p.5

```
<< LinearAlgebra`MatrixManipulation`
```

```
eqns = {x2 - 4 x3 == 8,
   2 x1 - 3 x2 + 2 x3 == 1,
   5 x1 - 8 x2 + 7 x3 == 1};

mv = LinearEquationsToMatrices[eqns, {x1, x2, x3}];
aug = makeAug[mv];
% // MatrixForm

RowReduce[aug] // MatrixForm
```

$$\begin{pmatrix} 0 & 1 & -4 & 8 \\ 2 & -3 & 2 & 1 \\ 5 & -8 & 7 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & -5 & 0 \\ 0 & 1 & -4 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

■ Lay Example 1.1.2, p.8

```
<< LinearAlgebra`MatrixManipulation`
```

```
eqns = {x1 - 2 x2 + x3 == 0,
             2 x2 - 8 x3 == 8,
   -4 x1 + 5 x2 + 9 x3 == -9};

mv = LinearEquationsToMatrices[eqns, {x1, x2, x3}];
aug = makeAug[mv];
% // MatrixForm

RowReduce[aug] // MatrixForm
```

$$\begin{pmatrix} 1 & -2 & 1 & 0 \\ 0 & 2 & -8 & 8 \\ -4 & 5 & 9 & -9 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & 29 \\ 0 & 1 & 0 & 16 \\ 0 & 0 & 1 & 3 \end{pmatrix}$$

■ Lay Example 1.1.3, p.9

```
<< LinearAlgebra`MatrixManipulation`
```

```
eqns = { x2 - 4 x3 == 8,
   2 x1 - 3 x2 + 2 x3 == 1,
   5 x1 - 8 x2 + 7 x3 == 1};

mv = LinearEquationsToMatrices[eqns, {x1, x2, x3}];
aug = makeAug[mv];
% // MatrixForm

RowReduce[aug] // MatrixForm
```

$$\begin{pmatrix} 0 & 1 & -4 & 8 \\ 2 & -3 & 2 & 1 \\ 5 & -8 & 7 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & -5 & 0 \\ 0 & 1 & -4 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Solving a System of Equations with LinearSolve

We wish to solve the linear system represented by the following augmented matrix (See Lay Example 1.1.1, p.4 ff.)

$$a = \begin{pmatrix} 1 & -2 & 1 & 0 \\ 0 & 2 & -8 & 8 \\ -4 & 5 & 9 & -9 \end{pmatrix};$$

LinearSolve provides an easy way to do that .
First, extract the relevant components, m and b.

```
m = TakeColumns[a, 3];
% // MatrixForm

b = TakeColumns[a, -1];
% // MatrixForm
```

$$\begin{pmatrix} 1 & -2 & 1 \\ 0 & 2 & -8 \\ -4 & 5 & 9 \end{pmatrix}$$

$$\begin{pmatrix} 0 \\ 8 \\ -9 \end{pmatrix}$$

Then use LinearSolve on the components.

```
s = LinearSolve[m, b]

{{29}, {16}, {3}}
```

Let's check that result.

```
m.s == b

True
```